

Se il software dei lettori del numero scorso non è per nulla bastato a calmare i vostri bollori smanetterecci che da bravi utenti Amiga certamente non vi mancheranno, questo mese vi «massacreremo» con una seconda lezione, sempre opera dello stesso autore, sul tema dello «scrolling & affini». Nella lettera che accompagnava l'articolo in questione, il prode Maurizio Mangrella ci consolava dicendo: «... quando, seguendo il filone della ricerca, ci si avvicina al mondo nascosto dei chip, scoprire nuove cose diviene sempre più difficile: nemmeno la grazia (!) di un Guru Meditation, ma solo blocchi inesorabili o allucinanti visioni che è meglio non descrivere...». Come lo capisco!

## Scrolling hardware, Copper et similia

di Maurizio Mangrella - Eboli (SA)

Domenica 12 giugno, ore 15: mentre il sole mi rosicchia lentamente il cervello, mi appresto ad uno dei più temibili viaggi nella memoria di un computer: vago nei meandri della grafica dell'Amiga. «Di nuovo?», si chiederà qualcuno. Sì, di nuovo, perché so che non ho visto tutto. Almeno non tutto quello che si poteva vedere.

Al caldo si aggiungono le cifre e le Guru Meditation, ed ho l'impressione di uscire fuori di senno, se non fosse per qualche risultato concreto che rinfranca lo spirito...

È inutile negarlo, Paolo Russo, l'autore di «B... come Blitter», mi ha lasciato perplesso: è possibile che in Basic non sia possibile «desiderare» i modi grafici speciali o lo scrolling hardware dello schermo? Al primo dilemma ho già dato risposta, ma il secondo continuava a non farmi dormire la notte.

Però qualcosa l'ho fatta, e ve la presento: anche da Basic, signori miei, è possibile scrollare — quant'è brutto questo! — la bitmap ed il video a livello hardware. La cosa è un tantino complessa, perciò vi invito a seguirmi attentamente.

Come già sappiamo, lo Screen Record — quella struttura dati che defini-

sce le caratteristiche di uno schermo — contiene altre due strutture, la ViewPort e la RastPort: la prima per i modi di visualizzazione, la seconda per l'allocatione dei bitplane in memoria ed altre cosette.

### Qualche altra routine

In aggiunta a quelle che vedremo dopo, riportiamo nella tabella sottostante la descrizione di alcune utili routine della intuition.library.

Per «puntare» qui si intende puntare al relativo record descrittore: così screen è il puntatore allo Screen Record e window quello al Window Record. Una routine che non chiede parametri in input non può essere chiamata da Basic come si farebbe da C (senza scrivere nulla tra le parentesi), ma bisogna fornire loro un «dummy argument», perfettamente inutile: ad esempio

```
vpos& = VBeamPos& (0)
```

### La ViewPort

La ViewPort è più importante di quanto si possa pensare — e di quanto io abbia pensato —: il suo schema è in figura 1. Prima di parlarne, ricordo che l'indirizzo della ViewPort contenente la Current Output Window è dato da

```
VPort& = 44+PEEK(L WINDOW 7 + 46)
```

o dalla routine ViewPortAddress della intuition.library, nella forma

### Alcune routine della intuition.library

#### ScreenToBack (screen)

Manda lo schermo puntato da screen dietro tutti gli altri

#### ScreenToFront (screen)

Porta lo schermo puntato da screen davanti a tutti gli altri

#### WindowToBack (window)

Come ScreenToBack, ma per la finestra puntata da window

#### WindowToFront (window)

Come ScreenToFront, ma per la finestra puntata da window

#### MoveScreen (screen, dx, dy)

Muove lo schermo puntato da screen di dx pixel in orizzontale (non illudetevi! non ha effetto) e di dy in verticale

#### MoveWindow (window, dx, dy)

Come MoveScreen, ma per la finestra puntata da window

#### DisplayBeep (screen)

Attua un «beep» visivo (inverte per un attimo il colore 0) sullo schermo puntato da screen

#### CloseWorkBench ()

Chiude tutte le finestre del WorkBench, conservandone in memoria solo i dati; se non ci sono altre finestre aperte chiude

anche lo schermo

#### OpenWorkBench ()

Riapre il WorkBench

Aggiungo anche tre routine della graphics.library:

#### GetRGB4 (colormap, i)

Ritorna una word nel formato XXXRRRRGGGGBBBB rappresentante le componenti cromatiche dell'i-esimo colore nella ColorMap puntata da colormap. Va usata nella forma  
DECLARE FUNCTION GetRGB4% LIBRARY  
LIBRARY "graphics.library"

```
c% = GetRGB4% (cm&,i&)
```

pena un "Overflow Error"

#### ScrollRaster (rastport, dx, dy, x1, y1, x2, y2)

Scrolla la BitMap della rastport puntata da rastport di dx pixels in orizzontale e dy in verticale, nell'ambito del rettangolo (x1, y1)-(x2, y2).

#### VBeamPos ()

Ritorna la posizione verticale del raster nel momento in cui viene chiamata.

```

struct ViewPort
{
    struct ViewPort *Next;
    struct ColorMap *ColorMap;
    struct CopList *DspIns;
    struct CopList *SprIns;
    struct CopList *ClrIns;
    struct UCopList *UCopIns;
    SHORT DWidth, DHeight;
    SHORT DxOffset, DyOffset;
    UWORD Modes;
    UBYTE SpritePriorities;
    UBYTE reserved;
    struct RasInfo *RasInfo;
}

```

Figura 1 - La ViewPort Structure.

VPort& = ViewPortAddress& (WINDOW 7)

Delle CopLists parleremo dopo, per il momento accontentiamoci — si fa per dire — del resto.

DWidth e DHeight sono, rispettivamente, larghezza ed altezza dello schermo: fin qui nulla di nuovo. La novità sta in DxOffset e DyOffset: queste due word definiscono la posizione del relativo schermo nell'ambito della videata. I loro indirizzi sono, rispettivamente, vport&+28 e vport&+30, dove port& è l'indirizzo della ViewPort. Già vi vedo tutti intenti a pokare in queste locazioni nella speranza di vedere l'agognato effetto; peccato, però, non succede niente... Aspettate il prossimo paragrafo e vedrete...

Poi c'è un byte dedicato alla priorità degli sprite: 8 bit per 8 sprite, dunque un bit per ogni sprite — sia lo schema che il funzionamento sono simili a quelli del 64 —; se un bit è settato, il relativo sprite sarà completamente sovrapposto alla BitMap, altrimenti sarà sovrapposto solo al colore 0 (e passerà sotto tutti gli altri). Per chi non lo sapesse, gli sprite hanno due bitplane e si condividono i registri colore nel seguente modo:

Colore mappato	Sprites: 0 - 1	2 - 3	4 - 5	6 - 7
0		trasparente		
1	registri 17	21	25	29
2	registri 18	22	26	30
3	registri 19	23	27	31

Per qualche oscuro motivo, allo sprite 0 — che è il puntatore del mouse, riservato al sistema operativo — corrisponde il bit 5...

Segue un byte riservato — che è meglio non modificare — ed il puntatore alla struct RasInfo, che «informa» la ViewPort sull'allocatione della BitMap in memoria e sul video. La conformazione di RasInfo è in figura 2.

RasInfo punta al prossimo RasInfo — questo schema «a rilancio» è tipico dell'Amiga — e l'ultimo non punta a un bel niente... Segue il puntatore alla BitMap — identico a quello contenuto nella RastPort — e i valori RxOffset e RyOffset, che servono a scrollare la BitMap all'interno dello schermo, lasciando inalterata la posizione di quest'ultimo rispetto al video. In pratica, questi due valori non sono altro che le coordinate del punto della BitMap che apparirà nell'angolo superiore sinistro dello schermo. Chiaro il giochetto?

```

struct RasInfo
{
    struct RasInfo *Next;
    struct BitMap *BitMap;
    SHORT RxOffset, RyOffset;
}

```

Figura 2 - La RasInfo Structure.

```

struct View
{
    struct ViewPort *ViewPort;
    struct cprlist *LOFCprList;
    struct cprlist *SHFCprList;
    SHORT DyOffset, DxOffset;
    UWORD Modes;
}

```

Figura 3 - La View Structure.

## View

Per chi mastica un po' di C amighese — è un dialetto particolare — il titolo sibillino di questo paragrafo dovrebbe dire tanto. Chiariamo per i nuovi arrivati.

Ciò che noi vediamo sul monitor del nostro Amiga è una View, la quale è divisa in «fette» orizzontali che si chiamano ViewPorts — ma guarda un po'...

È chiaro che il computer deve sapere come è organizzata la View: a questo scopo provvede una struttura dati, organizzata come recita la figura 3.

Il primo puntatore punta — lo dice il nome stesso... — alla prima ViewPort, che punta alla seconda, che punta alla terza... e così via. Poi vengono due voci di cui poche persone al mondo — a mio parere — conoscono il significato, dunque posizione e modo grafico del primo schermo.

Per cambiare i parametri di schermo non basta pokare nella ViewPort, ma bisogna anche ricalcolare la schermata; dunque, dopo aver settato i valori desiderati, dovremo dare (da Basic):

```

CALL MakeVPort (view&,vport&)
CALL MrgCop (view&)
CALL LoadView (view&)

```

Mamma li Turchi! Calma, calma, sono tutte routine della graphics.library; la prima ricostruisce la schermata puntata da view& — indirizzo di struct View — aggiungendo, nel caso ve ne fosse bisogno, la ViewPort puntata da vport&; la seconda (Merge CopLists) fonde le CopLists degli schermi di cui si compone la View e la terza spedisce il tutto a Denise.

## L'indirizzo di View

Ma dove sta View? Per capirlo ci ho messo parecchio, ma ora lo so — suspense cattiva atta a destare l'invidia dei lettori... Scherzo...

La View attualmente visualizzata sullo schermo — da ora in poi diremo «attiva» — è puntata da una locazione della parte bassa della RAM, a testimonianza che questi dati il S.O. li ha a sua disposizione — e se li tiene per sé.

Per sapere qual è il puntatore a View, dobbiamo aprire la graphics.library tra-

mite Exec — e non attraverso l'interprete Basic — dando:

GfxBase& — lo ricordate? In Assembler bisogna porlo nel registro A6, pena

una colorita Guru Meditation — punta ad una complessa struttura dati (figura 4) che contiene informazioni necessarie per il regolare svolgimento delle operazioni grafiche: per farla breve, l'indirizzo della View attiva è:

View& = PEEKL (Gf xBase&+34)

Lo 0, secondo parametro della OpenLibrary, indica al S.O. che non abbiamo particolari predilezioni circa la versione della relativa libreria: per noi vanno tutte bene. Per garantire la massima compatibilità, è bene specificare sempre 0.

Come al solito, esiste un metodo più semplice, basta dare:

View& = ViewAddress& (0)

```

DECLARE FUNCTION OpenLibrary& LIBRARY
LIBRARY "exec.library"

.
.
.

GfxBase& = OpenLibrary& (SADD("graphics.library"+CHR$(0)),0)

In C daremmo

#include <graphics/gfxbase.h>

.
.
.

GfxBase = (struct GfxBase *)OpenLibrary ("graphics.library",0)
    
```

```

struct GfxBase
(
    struct    Library LibNode;
    struct    View *ActiView;
    struct    copinit *copinit;
    LONG      *cia;
    LONG      *blitter;
    UWORD     *LOFlist;
    UWORD     *SHFlist;
    struct    bltnode *blthd,*blttl;
    struct    bltnode *bsblthd,*bsblttl;
    struct    Interrupt vbsrv,timsrv,bltsrv;
    struct    List TextFonts;
    struct    TextFont *DefaultFont;
    UWORD     Modes;
    BYTE      VBlank;
    BYTE      Debug;
    SHORT     BeamSync;
    SHORT     system_bplcon0;
    UBYTE     SpriteReserved;

    UBYTE     bytereserved;
    USHORT    Flags;
    SHORT     BlitLock;
    SHORT     BlitNest;
    struct    List BlitWaitQ;
    struct    Task *BlitOwner;
    struct    List TOF_WaitQ;
    UWORD     DisplayFlags;
    struct    SimpleSprite **SimpleSprites;
    UWORD     MaxDisplayRow;
    UWORD     MaxDisplayColumn;
    UWORD     NormalDisplayRows;
    UWORD     NormalDisplayColumns;
    UWORD     NormalDPMX;
    UWORD     NormalDPMY;
    struct    SignalSemaphore *LastChanceMemory;
    UWORD     *LCMptr;
    UWORD     MicrosPerLine;
    ULONG     reserved[2];
)
    
```

Figura 4 - La GfxBase Structure.




Figura 5  
L'albero genealogico delle strutture grafiche.

(la ViewAddress () è una routine della intuition.library). In Basic usare l'uno o l'altro metodo è quasi indifferente, mentre da C, paradossalmente, conviene il primo, in quanto ci evita di dover aprire anche la intuition.library.

## Il Copper

Bene bene, è venuta ora di parlare un po' del nostro beneamato Copper.

Il Copper — abbreviazione di coprocessor — è una parte del chip Agnus che si preoccupa di risistemare i puntatori alle zone grafiche utilizzati da Denise: in pratica, i registri del chip grafico vengono gestiti quasi esclusivamente dal Copper. Ad esempio, i registri DFF0E0 - DFF0F7, che contengono gli indirizzi dei bitplane, sono manipolabili solo dal Copper, ed un qualunque tentativo di scrittura attuato dal microprocessore non ha effetto.

Il Copper è, in realtà, una sorta di piccolo microprocessore RISC dedicato, che riesce ad eseguire solo tre istruzioni diverse — a quanto so — basandosi prevalentemente sulla posizione del pannello elettronico.

Ad ogni cambio di schermo — quando termina la parte visibile di una ViewPort e comincia quella successiva — il nostro coprocessore sistema tutti i puntatori: lavoro ingrato, questo, che gli costa circa due o tre righe vuote, ovvero due o tre Horizontal Blanks (tra 128 e 192 microsecondi, tempo in cui nulla, nemmeno gli sprite, è visualizzabile).

Le CopLists — e così chiamiamo ogni dubbio — sono i programmi del Copper, intesi come elenchi di istruzioni. Una istruzione presenta tipicamente il seguente formato:

```
struct CopIns
{
    SHORT    OpCode;
    LONG     *NextList;
    SHORT    VWaitPos;
    SHORT    DestAddr;
    SHORT    HWaitPos;
    SHORT    DestData;
}
```

(ho cercato di semplificare un po': il codice originale presenta delle union).

Nulla so degli OpCodes, so solo che sono tre — i nomi dovrebbero essere MOVE, SEARCH e SKIP, come lessi tempo fa su una rivista, ma non so se e quanto ciò sia vero —; una cosa è degna di nota: il Copper è talmente veloce da riuscire ad intercettare anche il percorso orizzontale del pannello elettronico: questo è indubbiamente vantaggioso, se è vero che la SEARCH può generare un interrupt per il 68000 e

```
' Viaggio all'interno dell'Amiga c by Maurizio Mangrella 1988
DECLARE FUNCTION OpenLibrary& LIBRARY ' Apriamo la graphics.library
LIBRARY "exec.library" ' tramite Exec
LIBRARY "graphics.library"

' Si parte !
GfxBase& = OpenLibrary&(5ADD("graphics.library"+CHR$(0)),0)
View& = PEEKL(GfxBase&+34) ' Indirizzo di ActiView
VPort& = 44+PEEKL(WINDOW(7)+45) ' La ViewPort del Wb Screen
RasInfo& = PEEKL(VPort&+35) ' Puntatore a RasInfo
RxOffset& = RasInfo&+8 ' Indirizzo di RxOffset
RyOffset& = RasInfo&+10 ' e di RyOffset
x = PEEKW(RxOffset&) ' Posizioni iniziali
y = PEEKW(RyOffset&)
WHILE 1 ' Comincia il viaggio (eterno!)
    a$ = ""
    WHILE a$ = "" : a$ = INKEY$ : WEND ' Un tasto cursore, please ...
    a = ASC(a$)
    IF a = 28 THEN y = y+1 ' Sale
    IF a = 29 THEN y = y-1 ' Scende
    IF a = 30 THEN x = x-1 ' A destra
    IF a = 31 THEN x = x+1 ' E a sinistra
    POKEW RxOffset&,x ' Settiamo i nuovi valori ...
    POKEW RyOffset&,y
    CALL MakeVPort(View&,VPort&) ' ... e li mettiamo in pratica
    CALL MrgCop(View&)
    CALL LoadView(View&)
WEND ' Continua
```

Figura 6 - Con i tasti cursore viaggerete nella memoria dell'Amiga...

```
' Viaggio sul monitor dell'Amiga c by Maurizio Mangrella 1988
' (In pratica e' lo stesso di prima ...)
DECLARE FUNCTION OpenLibrary& LIBRARY ' Apriamo la graphics.library
LIBRARY "exec.library" ' tramite Exec
LIBRARY "graphics.library"

' Si parte !
GfxBase& = OpenLibrary&(5ADD("graphics.library"+CHR$(0)),0)
View& = PEEKL(GfxBase&+34) ' Indirizzo di ActiView
VPort& = 44+PEEKL(WINDOW(7)+45) ' La ViewPort del Wb Screen
DxOffset& = VPort&+28 ' Indirizzo di DxOffset
DyOffset& = VPort&+30 ' e di DyOffset
sx = PEEKW(DxOffset&) ' Posizioni iniziali
sy = PEEKW(DyOffset&)
x = sx : y = sy
dx = -2 : dy = -2
WHILE 1 ' Comincia il rimbalzo
    x = x+dx ' Il solito algoritmo ...
    y = y+dy
    IF x < (sx-30) OR x >= (sx+30) THEN dx = -dx
    IF y < (sy-20) OR y >= (sy+20) THEN dy = -dy
    POKEW DxOffset&,x ' Settiamo i nuovi valori ...
    POKEW DyOffset&,y
    CALL MakeVPort(View&,VPort&) ' ... e li mettiamo in pratica
    CALL MrgCop(View&)
    CALL LoadView(View&)
WEND ' Continua
```

Figura 7 - Lo schermo che rimbalza sul monitor.

```
' WorkBench 320x200 e sprite diafani ! c by Maurizio Mangrella 1988
' (Ho inserito solo i commenti necessari)
DECLARE FUNCTION OpenLibrary& LIBRARY
LIBRARY "exec.library"
LIBRARY "graphics.library"
GfxBase& = OpenLibrary&(5ADD("graphics.library"+CHR$(0)),0)
View& = PEEKL(GfxBase&+34)
VPort& = 44+PEEKL(WINDOW(7)+45)
POKEW VPort&+32,0 ' ViewPort Mode (0 = LORES)
POKE VPort&+34,0 ' SpritePriorities
CALL MakeVPort(View&,VPort&)
CALL MrgCop(View&)
CALL LoadView(View&)
CALL CloseLibrary(GfxBase&) ' Chiudiamo la graphics.library
LIBRARY CLOSE ' Chiudiamo le librerie
```

Figura 8 - Workbench 320x200 e sprite a priorità più bassa.

```

' OverScan : una realta' anche da BASIC !      c Maurizio Mangrella 1988
DECLARE FUNCTION OpenLibrary& LIBRARY
LIBRARY "exec.library"
LIBRARY "graphics.library"
GfxBase& = OpenLibrary&(SADD("graphics.library"+CHR$(0)),0)
View& = PEEKL(GfxBase&+34)      ' Risparmiamo la intuition.library
SCREEN 1,352,240,5,1          ' OverScan 352x240
WINDOW 2,"Prova OverScan",,0,1  ' Aprire una finestra solo per i pa -
WINDOW OUTPUT 2              ' rametri di schermo, in fondo, e' pur
VPort& = 44+PEEKL(WINDOW(7)+46)  ' sempre il metodo piu' sbrigativo ...
POKEW VPort&+28,-16          ' 16 pixels a sinistra
CALL MakeVPort(View&,VPort&)    ' il solito Trio Lescano
CALL MrgCop(View&)
CALL LoadView(View&)
RANDOMIZE TIMER
FOR k = 1 TO 200              ' Linee casuali
  c = FIX(32*RND)
  x1 = FIX(352*RND)
  x2 = FIX(352*RND)
  y1 = FIX(240*RND)
  y2 = FIX(240*RND)
  LINE (x1,y1)-(x2,y2),c
NEXT k
WHILE INKEY$ = "" : WEND      ' Premere un tasto ...
CLS
FOR k = 1 TO 100              ' Rettangoli a caso
  c = FIX(32*RND)
  x1 = FIX(352*RND)
  x2 = FIX(352*RND)
  y1 = FIX(240*RND)
  y2 = FIX(240*RND)
  LINE (x1,y1)-(x2,y2),c,bf
NEXT k
WHILE INKEY$ = "" : WEND      ' Premere un tasto ...
WINDOW CLOSE 2
SCREEN CLOSE 1
CALL CloseLibrary(GfxBase&)
LIBRARY CLOSE

```

Figura 9 - OverScan da Basic: 352x240 in 32 colori.

che, dunque, sarebbe possibile cambiare parametri di schermo molte volte in una sola riga.

Come da figura 1, una ViewPort contiene tre CopLists — non so cosa sia UCop — Ins — Dsplns (Display Instructions, programma per il display), Sprlns (Sprite Instructions, programma per gli sprite) e Crlns (???)

Quando si uniscono più ViewPorts in

### IntuiMessage

Lo so, lo so, non c'entra (quasi) niente, ma, dato che l'ho scoperto proprio adesso, mentre sto scrivendo quest'articolo (meraviglie del multitasking), voglio inserirlo lo stesso: ecco come si fa ad ottenere un messaggio dal port di Intuition.

L'indirizzo del port legato alla Current Output Window è dato da: ▼

```
PEEKL (WINDOW(7)+85)
```

Per ottenere un messaggio (da BASIC), daremo

```
DECLARE FUNCTION GetMsg& LIBRARY
LIBRARY "exec.library"
```

```

*
*

```

```
Port& = PEEKL (WINDOW(7)+85)
Msg& = GetMsg& (Port&)
```

una View, le CopLists devono essere fuse insieme: di ciò si occupa MrgCop.

Ad ogni Vertical Blank il Copper invia un interrupt al 68000, che viene intercettato tramite la WaitTOF () (Wait for Top Of Frame).

Fatte le debite semplificazioni, la struct IntuiMessage (il cui puntatore è, nell'esempio precedente, Msg&) è quella mostrata in figura 10.

C'è un Node (informazioni per i tecnici: niente nome, priorità 0, tipo 5, corri-

```

struct IntuiMessage
{
  struct Message ExecMessage;
  ULONG Class;
  USHORT Code;
  USHORT Qualifier;
  APTR IAddress;
  SHORT MouseX, mouseY;
  ULONG Seconds, Micros;
  struct Window *IDCMPWindow;
  struct IntuiMessage *SpecialLink;
};

struct Message
{
  struct Node mn_Node;
  struct MsgPort *mn_ReplyPort;
  UWORD mn_Length;
};

struct Node
{
  struct Node *ln_Succ;
  struct Node *ln_Pred;
  UBYTE ln_Type;
  BYTE ln_Pri;
  char *ln_Name;
}

```

Figura 10 - La IntuiMessage Structure.

spondente a MESSAGE), il puntatore al ReplyPort (il port cui ripetere il messaggio), la lunghezza del messaggio e gli «items» che abbiamo visto l'altra volta: Class e Code, insieme ad un non meglio specificato Qualifier, poi (tante grazie!) ascissa e ordinata del mouse e secondi e microsecondi trascorsi dall'accensione... Dunque avremo:

```

Class& = PEEKL (Msg&+20)
Code% = PEEKW (Msg&+24)
Qualifier% = PEEKW (Msg&+26)
MouseX% = PEEKW (Msg&+32)
MouseY% = PEEKW (Msg&+34)
Seconds& = PEEKL (Msg&+36)
Micros& = PEEKL (Msg&+40)

```

Bel colpo!

### Conclusioni

Qui termina il mio lavoro, come al solito aperto ad ampliamenti da parte di chi sia riuscito a scoprire qualcosa di più. Così fughiamo ogni dubbio anche per quanto riguarda il Basic: con questo linguaggio è possibile quasi tutto; non si può certo pretendere di realizzare un'animazione tipo Juggler o di riuscire ad ottenere effetti che solo il Linguaggio Macchina consente (e solo in casi particolari). Qualcosa di buono si può comunque fare...

A presto!



